

Vježba 2: Koeficijent apsorpcije

<http://sail.zpf.fer.hr/laserlab/beamabs.nb>

1. Unos podataka

Unose se parovi $\{x_i, P_i\}$ gdje je x_i u cm, a P_i u mW:


```
In[1]:= data = {{2, 0.38}, {2.5, 0.355}, {5, 0.24}, {6, 0.2}, {8, 0.14}}
```

```
Out[1]= {{2, 0.38}, {2.5, 0.355}, {5, 0.24}, {6, 0.2}, {8, 0.14}}
```

```
In[2]:= n = Length@data
```

```
Out[2]= 5
```

```
In[3]:= datafig = {{Red, AbsolutePointSize[5], Point /@ data}};
Show[Graphics@datafig, Axes -> True, AxesLabel -> {"x [cm]", "P [mW]"}];
```



2. Linearizirani model: $y_i = \ln P_i = \ln P_0 - \alpha x_i = a + b x_i$

Pogledati upute 2.4.1 i A.5

Funkcija koja racuna redak tablice:

```
In[5]:= redakfun[{x_, p_}] := Module[{y = Log[p]}, {x, p, y, x^2, x y, y^2}]
```

```
In[6]:= (tablica = redakfun /@ data) // TableForm
```

```
Out[6]//TableForm=
```

2	0.38	-0.967584	4	-1.93517	0.936219
2.5	0.355	-1.03564	6.25	-2.58909	1.07255
5	0.24	-1.42712	25	-7.13558	2.03666
6	0.2	-1.60944	36	-9.65663	2.59029
8	0.14	-1.96611	64	-15.7289	3.8656

Zbroj vrijednosti u stupcima tablice:

```
In[7]:= {sx, tmp, sy, sx2, sxy, sy2} = Total /@Transpose@tablica
```

```
Out[7]= {23.5, 1.315, -7.00589, 135.25, -37.0454, 10.5013}
```

Velicina Δ prema izrazu (A.19):

```
In[8]:= del = n sx2 - sx2
```

```
Out[8]= 124.
```

Ocekivana vrijednost parametra a prema izrazu (A.17):

```
In[9]:= a = (sx2 sy - sx sxy) / del
```

```
Out[9]= -0.620808
```

Ocekivana vrijednost parametra b prema izrazu (A.18):

```
In[10]:= b = (n sxy - sx sy) / del
```

```
Out[10]= -0.166036
```

Standardna pogreska parametra b prema izrazu (A.23):

```
In[11]:= db = Sqrt [  $\frac{n}{del (n - 2)}$  (sy2 - a sy - b sxy) ]
```

```
Out[11]= 0.00390281
```

Koeficijent apsorpcije je, u lineariziranom modelu, konacno:

```
In[12]:= alphafitlin = -b ± db
```


```
Out[12]= 0.166036 ± 0.00390281
```

Rel. pogreska:

```
In[13]:= 100 Abs [db / b]
```

```
Out[13]= 2.35058
```

```
In[14]:= Plot [Exp [a + b x], {x, 0, (3 / 2) Max@First@Transpose@data},  
  PlotRange → {-0.05 Exp@a, 1.05 Exp@a},  
  AxesLabel → {"x [cm]", "P [mW]"},  
  Epilog → datafig];
```



3. Nelinearni model: $P_i = P_0 \text{Exp}[-\alpha x_i]$

Pogledati upute 2.4.2 i A.4

■ 3.1 Prilagodbeni postupak (fit)

Za pretpostavljenu vrijednost α funkcija p0fun racuna ocekivanu vrijednost P_0 prema formuli (2.12) iz uputa:


```
In[15]:= p0fun[α_, data_] := Module[{xi, pi, tmp},
  {xi, pi} = Transpose@data;
  tmp = Exp[-α xi];
  pi.tmp / tmp.tmp
]
```

Funkcija s2fun racuna S^2 prema formuli (2.11) uz vrijednost P_0 dobivenu funkcijom p0fun:

```
In[16]:= s2fun[α_, data_] := Module[{xi, pi, tmp},
  {xi, pi} = Transpose@data;
  tmp = pi - p0fun[α, data] Exp[-α xi];
  tmp.tmp
]
```

Crtamo S^2 u ovisnosti o parametru α :

```
In[17]:= Plot[s2fun[α, data], {α, 0, 0.5}, AxesLabel → {"α", "S²"}, PlotPoints → 64];
```



Pronalazimo minimum S^2 u odnosu na parametar α :

```
In[18]:= s2min = FindMinimum[s2fun[α, data], {α, 0, 0.5}]
```


```
Out[18]= {0.0000553051, {α → 0.162604}}
```

Ocekivane vrijednosti parametara:

```
In[19]:= {alphahat, p0hat} = {α, p0fun[α, data]} /. Last[s2min]
```

```
Out[19]= {0.162604, 0.530278}
```

```
In[20]:= Plot[p0hat Exp[-alphahat x], {x, 0, (3/2) Max@First@Transpose@data},
  PlotRange → {-0.05 p0hat, 1.05 p0hat},
  AxesLabel → {"x [cm]", "P [mW]"},
  Epilog → datafig];
```



■ 3.2 Procjena pogreske (nije opisano u uputama)

Napomena: postupak procjene pogreske nije opisan u uputama.

Definira se reducirani χ^2 :

```
In[21]:= chi2redfun[α_, data_] = (n - 2) s2fun[α, data] / s2fun[alphahat, data];
```

Druga derivacija po α reduciranog χ^2 u minimumu:

```
In[22]:= fdd = D[chi2redfun[α, data], {α, 2}] /. {α → alphahat}
```

```
Out[22]= 135769.
```


Standardna (χ^2 -curvature) pogreska parametra α :

```
In[23]:= sigalph = 2 / fdd // Sqrt
```

```
Out[23]= 0.00383808
```

Kontrolni graf:

```
In[24]:= Plot[(n - 2) s2fun[α, data] / First[s2min],
  {α, alphahat - 1.5 sigalph, alphahat + 1.5 sigalph}, FrameLabel → {"α", "S²"},
  PlotRange → {n - 2 - 0.5, n - 2 + 1.5}, Frame → True, Axes → None,
  GridLines → {{alphahat, alphahat - sigalph, alphahat + sigalph}, {n - 2, n - 2 + 1}}];
```



Konacno apsorpcijski koeficijent:

```
In[25]:= alphafitnonlin = alphahat ± sigalph
```

```
Out[25]= 0.162604 ± 0.00383808
```

Rel. pogreska:

```
In[26]:= 100 Abs[sigalph / alphahat]
```

```
Out[26]= 2.36038
```